

The `roundrect` Macros, v2.0

Donald P. Goodman III

August 1, 2015

Abstract

The `roundrect` macros for METAPOST provide extremely configurable, extremely versatile rectangles (including rounded corners), intended primarily for inclusion in documents produced by T_EX and friends. The idea was to provide a METAPOST-based replacement for the incredibly versatile `tcolorbox` package; the macros are far from achieving that goal. But they are nevertheless extremely useful.

Contents

1	Introduction	1
2	Prerequisites and Conventions	2
3	Basic Usage	2
4	Implementation	8

1 Introduction

While T_ikZ and its many accompanying packages, particularly `tcolorbox`, are wonderful and powerful tools, whenever using them I inevitably feel completely lost, and I exert great effort doing comparatively simple things. Contrariwise, thanks to my experience with the `drm` and `dozenal` packages, writing in METAPOST is quite straightforward for me. So I decided to try to write some generalized macros to provide functionality similar to that of `tcolorbox`. It's not even close to that kind of flexibility or power, but it's still quite useful and versatile, so I make it available for anyone who might be interested.

This document was typeset in accordance with the `docstrip` utility, which allows the automatic extraction of code and documentation from the same document.

2 Prerequisites and Conventions


Some prerequisites for using this package are METAPOST itself (obviously). If you're using the package with L^AT_EX, the `gmp` package would probably be helpful; be sure to use the `latex` package option. Finally, the package internally calls `TEX.mp`, so that is also required. All of these should be packaged in any reasonably modern L^AT_EX system, such as T_EXLive or MikT_EX.

This documentation assumes nothing about your personal T_EX or METAPOST environment. ConT_EXt and the various forms of LuaT_EX have METAPOST built-in; with pdfL^AT_EX, the author's choice, one can use the `gmp` package to include the source directly in one's document (that's what's been done in this documentation) or develop a simple script to compile them afterwards and include them in the source via `\includegraphics` (probably the quickest option, since compilation is done in advance). Here, we simply post the plain vanilla METAPOST code, and let you work out those details however you prefer.

3 Basic Usage

roundrect The core of all the action is the `roundrect` macro; this will set up your rounded rectangle in the plainest way possible. The first argument is the box's height, the second its width, and the third its name, by which you will draw it later:

```
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



rrborderrad() All the corners don't *have* to be rounded; we can make them square if we want. To do things like this, we use the macro `rrborderrad()`, which takes a single argument giving the border radius we want; that is, how rounded we want the corners of our rectangle. Higher values will be more rounded, lower values will be less:

```
rrborderrad(10pt);  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



Notice that the corners in this, with `rrborderrad()` set to `10pt`, are much less rounded than the previous example. The default border radius is `40pt`, which is quite rounded.

`rrborderrad()` provides an easy way to set the border radius of all four corners at once; however, we can also control each corner individually, with `rrtoplftborderrad`, `rrbotlftborderrad`, `rrtoprtborderrad`, and `rrbotrtborderrad`, which are parameters rather than macros; that is, we define them using `:=` rather than as an argument in parentheses:

```
rrtoplftborderrad := 20pt;  
rrbotlftborderrad := 40pt;  
rrtoprtborderrad := 10pt;  
rrbotrtborderrad := 60pt;  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



As you can see, this makes it possible to create a large variety of shapes, including the ability to arbitrarily flatten any side of the rectangle desired just by setting the border radius of the appropriate corners to `0pt`:

```
rrtoplftborderrad := 0pt;  
rrtoprftborderrad := 0pt;  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



Here, we've flattened the top border by setting the top right and top left corners' border radii to `0pt`. This ability to flatten any given side of the rectangle makes it much easier to combine multiple rectangles into interesting forms, which we'll see a bit more about later.

Of course, the color of both the background and the border can be controlled with `rrinnercolor` and `rrbordercolor()`, respectively.

```
rrinnercolor  
rrbordercolor
```

```
rrbordercolor(blue);  
rrinnercolor := red;  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



By default, `rrinnercolor` is white and `rrbordercolor` is black. Notice that `rrbordercolor` is a *macro*, not a parameter; that's because each border can be individually colored, and this macro simply does all of them at once. We'll see more about that later.

```
rrnotop  
rrnobot  
rrnolft  
rrnort
```

You can also completely suppress the border by using `rrnotop`, `rrnobot`, `rrnolft`, and `rrnort`, which is particularly useful when you want to combine multiple rectangles without making an obvious border between them. You can combine these in any way you like:

```
rrbordercolor(blue);
rrinnercolor := red;
rrnotop := true;
rrnobot := true;
rrborderrad(0pt);
roundrect(1in,2in)(rectangle);
draw rectangle;
```



Here we've squared all the corners to make it easier to see what's going on.
Each border can be colored individually and separately from the others, using the commands you'd expect:

```
rrtopbordercolor := blue;
rrbotbordercolor := green;
rrlftbordercolor := red;
rrrtbordercolor := black;
rrborderrad(20pt);
roundrect(1in,2in)(rectangle);
draw rectangle;
```



There is obviously some difficulty in determining what part of each rounded corner should be colored how; this ability is typically more useful with a single, flattened side, to help it blend in better when combined with other constructs:

```
rrbordercolor(black);
rrbotbordercolor := green;
rrinnercolor := red;
rrborderrad(20pt);
rrbotlftborderrad := 0pt;
rrbotrtborderrad := 0pt;
roundrect(1in,2in)(rectangle);
draw rectangle;
```



Perhaps you don't like the border; you'd like it thicker, or drawn with a square rather than a circular pen. You're in luck; `rrborderpen()` takes the single argument of the pen you'd like to draw the border with, defined like any other METAPOST pen:

```
rrborderpen(pensquare scaled 3);  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



The default border pen is `pencircle scaled 1.5`, so this results in a square pen rather than a circular one, twice as thick. You can also use individual pens for each border, as expected:

```
rrbotlftborderrad := 0pt;  
rrbotrtborderrad := 0pt;  
rrbotbordercolor := green;  
rrbotborderpen := pensquare yscaled 6;  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```



Here we've flattened the bottom border, colored it green, and drawn it with a square pen scaled on the y-axis only by 6. Clearly, there are huge possibilities here.

Finally, we can put text in the rectangles; this is as configurable as everything else:

```
rrbodytext := "Let's put some text in  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```

Let's put some text into this rectangle and see if it typesets correctly!

rrtextfont The font and style of the text can be controlled with **rrtextfont**, and the
rrtextcolor color of the text can be controlled with **rrtextcolor**:

```
rrbodytext := "Text in a label";  
rrtextcolor := green;  
rrtextalign := "\raggedleft";  
rrtextfont := "\fontsize17pt19pt\ sel  
roundrect(1in,2in)(rectangle);  
draw rectangle;
```

Text in a label

rrtextalign We also used, without explaining it first, **rrtextalign**, which allows insertion of text alignment commands. This can also be inserted in the **rrtextfont** variable, but it seemed logical to have a separate parameter for it. It's default is **\centering**.

rrtextwd The width of the text is governed by **rrtextwd**, which defaults to the same width as the rectangle with a 3pt buffer on either side. The buffer is not directly controllable, but the width can be set however you like:

```

rrbodytext := "Let's put some text in";
rrtextwd := 80pt;
roundrect(1in,2in)(rectangle);
draw rectangle;

```

Let's put some
text into this
rectangle and see
if it typesets
correctly!

if it typesets correctly!

Finally, the restore all these values to the default, use the `rrrestorevals` directive. This will clear everything to default so you can have a completely different `roundrect` in the same figure.

4 Implementation

```

1 input TEX;
2 color rrinnercolor; rrinnercolor := white;
3 numeric rrtoprtborderrad; rrtoprtborderrad := 40pt;
4 numeric rrbotrtborderrad; rrbotrtborderrad := 40pt;
5 numeric rrbotlftborderrad; rrbotlftborderrad := 40pt;
6 numeric rrtoplftborderrad; rrtoplftborderrad := 40pt;
7 numeric rrtextwd; rrtextwd := 0;
8 string rrtextfont; rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";
9 color rrtextcolor; rrtextcolor := black;
10 string rrbodytext; rrbodytext := "";
11 string rrtextalign; rrtextalign := "\centering";
12 boolean rrnotop; rrnotop := false;
13 boolean rrnobot; rrnobot := false;
14 boolean rrnolft; rrnolft := false;
15 boolean rrnort; rrnort := false;
16 color rrtopbordercolor; rrtopbordercolor := black;
17 color rrbotbordercolor; rrbotbordercolor := black;
18 color rrlftbordercolor; rrlftbordercolor := black;
19 color rrrtbordercolor; rrrtbordercolor := black;
20 def rrbordercolor(expr x) =
21 rrtopbordercolor := x;
22 rrbotbordercolor := x;
23 rrlftbordercolor := x;
24 rrrtbordercolor := x;
25 enddef;
26 def rrborderrad(expr x) =
27 rrtoplftborderrad := x;
28 rrbotlftborderrad := x;

```



```

29 rrtoprtborderrad := x;
30 rrbotrtborderrad := x;
31 enddef;
32 pen rrtopborderpen; rrtopborderpen := pencircle scaled 1.5;
33 pen rrbotborderpen; rrbotborderpen := pencircle scaled 1.5;
34 pen rrlftborderpen; rrlftborderpen := pencircle scaled 1.5;
35 pen rrrtborderpen; rrrtborderpen := pencircle scaled 1.5;
36 def rrborderpen(expr x) =
37   rrtopborderpen := x;
38   rrbotborderpen := x;
39   rrlftborderpen := x;
40   rrrtborderpen := x;
41 enddef;
42 def rrrestorevals =
43   rrborderrad(40pt);
44   rrbordercolor(black);
45   rrborderpen(pencircle scaled 1.5);
46   rrinnercolor := white;
47   rrnotop := false;
48   rrnobot := false;
49   rrnolft := false;
50   rrnort := false;
51   rrtextwd := 0;
52   rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";
53   rrtextcolor := black;
54   rrbodytext := "";
55   rrtextalign; rrtextalign := "\centering";
56 enddef;
57 def roundrect(expr rrht, rrwd)(suffix name) =
58   TEXPRE("%&latex" & char(10) & "\documentclass{article}\begin{document}");
59   TEXPOST("\end{document}");
60   if (rrtextwd = 0):
61     rrtextwd := rrwd - 12pt;
62   fi
63   path rra; path rrb; path rrc; path rrd;
64   pair a; pair b; pair c; pair d;
65   a := (0,0) shifted (-rrwd/2,-rrht/2);
66   b := (0,0) shifted (rrwd/2,-rrht/2);
67   c := (0,0) shifted (rrwd/2,rrht/2);
68   d := (0,0) shifted (-rrwd/2,rrht/2);
69   rra := fullcircle scaled rrbotlftborderrad shifted (xpart a +
70     (rrbotlftborderrad/2),ypart a + (rrbotlftborderrad/2));
71   rrb := fullcircle scaled rrbotrtborderrad shifted (xpart b -
72     (rrbotrtborderrad/2),ypart b + (rrbotrtborderrad/2));
73   rrd := fullcircle scaled rrtoplftborderrad shifted (xpart d +
74     (rrtoplftborderrad/2),ypart d - (rrtoplftborderrad/2));
75   rrc := fullcircle scaled rrtoprtborderrad shifted (xpart c -
76     (rrtoprtborderrad/2),ypart c - (rrtoprtborderrad/2));
77   pair f; f := (a--b) intersectionpoint rra;
78   pair g; g := (a--b) intersectionpoint rrb;

```

```

79 pair h; h := (b--c) intersectionpoint rrb;
80 pair i; i := (b--c) intersectionpoint rrc;
81 pair j; j := (c--d) intersectionpoint rrc;
82 pair k; k := (c--d) intersectionpoint rrd;
83 pair l; l := (d--a) intersectionpoint rrd;
84 pair m; m := (d--a) intersectionpoint rra;
85 picture name;
86 picture border;
87 picture rrtext;
88 pair n; pair o;
89 path rrtoplftcorner; path rrbotlftcorner;
90 path rrtoprtcorner; path rrbotrtcorner;
91 path rrtopborder; path rrbotborder;
92 path rrlftborder; path rrrtborder;
93 rrtoplftcorner := l{up}..{right}k;
94 rrtoprtcorner := j{right}..{down}i;
95 rrbotrtcorner := h{down}..{left}g;
96 rrbotlftcorner := f{left}..{up}m;
97 rrtopborder := rrtoplftcorner--rrtoprtcorner;
98 rrbotborder := rrbotrtcorner--rrbotlftcorner;
99 rrlftborder := rrbotlftcorner--rrtoplftcorner;
100 rrrtborder := rrtoprtcorner--rrbotrtcorner;
101 name := image(fill rrtoplftcorner--rrtoprtcorner--
102 rrbotrtcorner--rrbotlftcorner--cycle withcolor
103 rrinnercolor);
104 picture rrtmpborder;
105 border := currentpicture;
106 if (rrnotop = false):
107 rrtmpborder := image(draw rrtopborder withcolor
108 rrtopbordercolor withpen rrtopborderpen);
109 addto border also rrtmpborder;
110 fi
111 if (rrnobot = false):
112 rrtmpborder := image(draw rrbotborder withcolor
113 rrbotbordercolor withpen rrbotborderpen);
114 addto border also rrtmpborder;
115 fi
116 if (rrnolft = false):
117 rrtmpborder := image(draw rrlftborder withcolor
118 rrlftbordercolor withpen rrlftborderpen);
119 addto border also rrtmpborder;
120 fi
121 if (rrnort = false):
122 rrtmpborder := image(draw rrrtborder withcolor
123 rrrtbordercolor withpen rrrtborderpen);
124 addto border also rrtmpborder;
125 fi
126 addto name also border;
127 rrtext :=
128 image(label(TEX("\parbox{"&decimal(rrtextwd)&"bp}{&rrtextalign&rrtextfont&" "&rrbodytext&"}"),

```

```
129 addto name also rrtex;  
130 enddef;
```